
ABSTRACT  
Using knowledge of the windows file naming and folder structure to rapidly process multiple data files using SAS

®
 

Macro code. 
 

INTRODUCTION  
When presented with a need to process many sets of similar data, it became apparent, that if you could use a 
filename structure that had meaning within the window folder structure it would make a convenient and rapid way to 
process data and create individual specially named reports.  Combining the folder / file naming abilities of windows 
and the SAS Macro code processing of variable / filename creations could be a perfect marriage. 
 
In the environment of my employment, there are 175+ sets of similar data to be processed on a regular schedule.  
Each file represents a district’s worth of information.  Each district has been assigned a unique code number so that 
each district’s data file can contain the district’s code. For example, let’s say that the unique code for a district named 
APPLE was 001, then the data file for that district could be named Data_001_Oct2005.txt   Where the 001 represents 
the district code number, the other parts of the file name provides descriptive information about its creation date or 
other descriptive information as needed.  Our current focus will be only on the unique district code contained in the file 
name. 
 
To make this an easy system for maintenance, the district code and the district name are maintained in a simple file 
structure, an Excel spreadsheet is used.  This file is in an available form for SAS to read for actual processing. 
 
Here is an overview of the process: 
 
Files to be processed are placed into a “staging folder”.   
 
In the SAS program 
 Read simple file structure that contains district code and name, and create a format of the  information. 

Read “staging folder” and determine how many files to process. 
 Loop through files in the “staging folder”, one at a time, read in filename to extract district code. 
 Using the extracted district code and format,  create a macro variable for the district name. 
 Read district data file and process information, create report. 
 Using macro variable of district name, and number, make a file name similar to the form: 
  REPORT_001_APPLEdistrict.PDF 
 Go back and get next file name, and process again. 
 
 

CONCLUSION  
The process described can be ported to other operating systems.  Whenever there is a need for repetitive processing, 
this process can be used in one form or another.  Keys are using the SAS Macro code language to generate 
meaningful variables that can be used in identifying the resulting reports and having an input filename structure that 
can identify the source. 

PROGRAM SOURCE 

libname mylib 'c:\SAStemp'; 
ods path mylib.tmplmst(read) sashelp.tmplmst(read); 
/******************************************************************************* 
/* Purpose of this routine 
/******************************************************************************* 
/*   To be able to automate the processing of creating reports for the multiple 
sources  
/******************************************************************************* 
/* 
/* Created by Edgar Adams  October 2005 
/* For MWSUG Cincinnati, Ohio 
/* 
/* Process is as follows.... 
/* 
/*   Read in Format statements for District / Number table 
/* 
/*   Create macro variables that contain 
/*      District Name 
/*      District Number 
/* 
/*   Read CSV file into a SAS dataset and do report 
/*   Use District Number and Name to label report 
/* 
 
*/ 
options Linesize = 80 
        Center 
        date 
        nonumber 
        ; 
 
Filename DIR 'C:\MWSUG\DATA\DistrictFiles\'; 
 
%Let RptDate = %sysfunc(date(), mmddyy6.); 
 
/*********** 
/*  Format for District Number and Names 
/*  Create District Name Format 
*/ 
 %Include "C:\MWSUG\PROGRAMS\DistrictFormat.txt"; 
 
/************ 
/* Begin processing of information by setting up the macros... 
/************ 
*/ 
ODS Listing Close; 
 
%macro CreateRpts(FileToRead); 
    /*Associate District Number to District Name for report writing */ 
    Data _Null_; 
     District_Number = &Dist; 
        strng = Trim(put(District_Number,Di_TBL.)); 
        call symput('DName', Trim(strng)); 
        strng2 = Trim(put(District_Number,z3.)); 
        call symput('dinum',Trim(Strng2)); 
    run; 
 
 
 
 
 



PROGRAM SOURCE 

/******************************************************************************/ 
/* Create individual district reports *****************************************/ 
/******************************************************************************/ 
 
ODS PDF style=oaartf 
file="C:\MWSUG\REPORT\Rpt_&dinum._%Trim(&DName)_&RptDate..PDF"; 
 
/*  Begin by build sas dataset from the CSV 
*/ 
data workfile; 
    %let _EFIERR_ = 0; /* set the ERROR detection macro variable */ 
    infile "C:\MWSUG\DATA\DistrictFiles\&FileToRead" 
            MISSOVER DSD lrecl=32767 firstobs=2 ; 
    informat District_Number  3. ; 
    informat Count     4. ; 
     
    format District_Number 3. ; 
    format Count           4. ; 
     
        input 
          District_Number  
          Count ; 
 
  if _ERROR_ then call symput('_EFIERR_',1);  /* set ERROR detection 
macro variable */ 
      run; 
 
   /*****************/ 
   /* Create report */ 
        /*****************/ 
         Proc Freq data= workfile; 
          title1 "Report"; 
          title2 "======"; 
          title3 "District: &DName"; 
            table Count / nocol nopercent norow; 
         run; 
 ODS PDF Close; 
%Mend;  *CreateRpts(FileToRead); 
 
%Macro CkDirectory(DirRef); 
        * Open directory to find out how many files to process; 
        %let drc = %sysfunc(dopen(&DirRef)); 
                %let NumFiles = %sysfunc(dnum(&drc)); 
                * Log entry for trouble shooting  :)  ; 
                %if (&NumFiles = .) %then %Put The directory is empty; 
                %else %Put The folder contains &NumFiles files; 
 
        %do I = 1 %to &NumFiles; 
                * Strip out District Number and Read filename; 
                %let Dist = %substr(%Sysfunc(dread(&drc,&I)),4,3); 
                %let LEFfileName = %Sysfunc(dread(&drc,&I)); 
                %CreateRpts(&LEFfileName); 
        %end; 
        %let rc=%sysfunc(dclose(&drc)); 
%mend;  * CkDirectory(DirRef); 
 
%CkDirectory(DIR); 



PROGRAM SOURCE  -  DISTRICT FORMAT 

        filename OfflList "C:\MWSUG\DATA\District_names_numbers_2004-2005.xls";             
                                                                                            
        proc import datafile=OfflList                                                       
            out=DistFILE                                                                    
            dbms=excel REPLACE;                                                             
        run;                                                                                
        DATA DistFORMAT; SET DistFILE;                                                      
            fMtname="Di_TBL";                                                              
            LABEL=DistrictNA;                                                               
            START=DistrictNO;                                                               
            END=DistrictNO;                                                                 
            KEEP FMTNAME LABEL START END;                                                  
        RUN;                                                                                
        PROC FORMAT CNTLIN=DistFORMAT;                                                      
        RUN;                                                                                
 
 
 
 
 
 
 
 
 
 
 



SAMPLE FORMAT DATA FILE 

 

 



SAMPLE DATA FILE 

 

 



SAS LOG  
 
Number of files to be processed 
 
 

 



SAMPLE SAS LOG – ONE DISTRICT PROCESSING 

 



SAMPLE REPORT 

